Task

Write a program that solves the odd word problem with the restrictions given below.



Description

You are promised an input stream consisting of English letters and punctuations.

It is guaranteed that:

 the words (sequence of consecutive letters) are delimited by one and only one punctuation,
 the stream will begin with a word,
 the words will be at least one letter long,   and 
 a full stop (a period, [.]) appears after, and only after, the last word.


Example

A stream with six words:

 what,is,the;meaning,of:life. 


The task is to reverse the letters in every other word while leaving punctuations intact, producing:

 what,si,the;gninaem,of:efil. 
while observing the following restrictions:

 Only I/O allowed is reading or writing one character at a time, which means: no reading in a string, no peeking ahead, no pushing characters back into the stream, and no storing characters in a global variable for later use;
 You are not to explicitly save characters in a collection data structure, such as arrays, strings, hash tables, etc, for later reversal;
 You are allowed to use recursions, closures, continuations, threads, co-routines, etc., even if their use implies the storage of multiple characters.


Test cases

Work on both the   "life"   example given above, and also the text:

 we,are;not,in,kansas;any,more.


import "io" for Stdin,Stdout
import "/str" for Char
 
var fwrite = Fn.new { |ch|
    System.write(ch)
    Stdout.flush()
}
 
var doChar // recursive
doChar = Fn.new { |odd, f|
    var c = Stdin.readByte()
    if (!c) return false // end of stream reached
    var ch = String.fromByte(c)
 
    var writeOut = Fn.new {
        fwrite.call(ch)
        if (f) f.call()
    }
 
    if (!odd) fwrite.call(ch)
    if (Char.isLetter(ch)) return doChar.call(odd, writeOut)
    if (odd) {
        if (f) f.call()
        fwrite.call(ch)
    }
    return ch != "."
}
 
for (i in 0..1) {
    var b = true
    while (doChar.call(!b, null)) b = !b
    Stdin.readByte() // remove \n from buffer
    System.print("\n")
}